query: SELECT ?item ?itemLabel ?image ?article ?sitelink
  WHERE 
  {
    {
      ?item wdt:P181 ?image.
      filter(strstarts(str(?image),"http://commons.wikimedia.org/wiki/Special:FilePath/Prostanthera"))
      ?sitelink ^schema:name ?article .
      ?article schema:about ?item ;
               schema:isPartOf <https://en.wikipedia.org/> .
    }
    MINUS
    {
      SERVICE wikibase:mwapi
      {
        bd:serviceParam wikibase:api "Search" .
        bd:serviceParam wikibase:endpoint "en.wikipedia.org" .
        bd:serviceParam mwapi:srnamespace "0" .
        bd:serviceParam mwapi:srlimit "max" .
        bd:serviceParam mwapi:srwhat "text" .
        bd:serviceParam mwapi:srsearch 'insource:"range_map = Prostanthera"' .
        ?title wikibase:apiOutput mwapi:title .
      }
      BIND (STRLANG(?title, "en") as ?sitelink)
    }
    SERVICE wikibase:label { bd:serviceParam wikibase:language "[AUTO_LANGUAGE],en". }
  }